

```

10 MODE 2
20 PRINT "          STATISTICS PROGRAM"
30 PRINT:PRINT "This program can be used to calculate"
40 PRINT:PRINT "means, moments, etc. for ungrouped and"
50 PRINT:PRINT "grouped data.":PRINT
60 PRINT:PRINT "ENTER 9999 when all data has been entered."
70 DG=1
80 PRINT:PRINT
90 INPUT"HIGHEST DATA";HD:PRINT
100 INPUT"LOWEST DATA";LD
110 ND=HD-LD+1
120 ND2=ND:DIV=1
130 WHILE ND>600
140 ND=ND2/DIV
150 DIV=DIV+1
160 WEND
170 IF DIV=1 THEN WIDE=600\ND2 ELSE WIDE=1
180 XLEN=(ND2*WIDE)/DIV
190 NUM=ND2/DIV+1
200 IF NUM<9 THEN NUM=NUM-1:DIV=NUM/10:WIDE=WIDE*NUM/10:NUM=10
210 DIM C(NUM)
220 CLS
230 PRINT:PRINT"Is data (1)ungrouped or (2)grouped?"
240 INPUT GR
250 IF GR=1 THEN 290
260 IF GR<>2 THEN 240
270 PRINT:INPUT"Group value";D:INPUT"Number in group";DG:GOTO 300
280 PRINT:INPUT"Group value";D:INPUT"Number in group";DG:GOTO 320
290 INPUT"DATUM";D
300 AT=D:GOTO 320
310 INPUT"DATUM";D
320 IF D=9999 THEN 420
330 IF D<LD OR D>HD THEN PRINT"OUT OF RANGE!":IF GR=1 THEN 310 ELSE 280
340 GL=LOG(D)
350 N=N+DG:LG=LG+(GL*DG):H=H+(DG/D)
360 TA=DG*(D-AT)+TA:TD=((D-AT)^2)*DG+TD
370 V=INT((D-LD)/DIV)+1
380 C(V)=C(V)+DG
390 IF C(V)>MOST THEN MOST=C(V)
400 IF GR=1 THEN 310 ELSE 280
410 REM ANALYSIS OF DATA
420 AM=AT+(TA/N):GM=EXP(LG/N):HM=N/H
430 CLS
440 PRINT:PRINT"ARITHMETIC MEAN =";AM
450 PRINT:PRINT"GEOMETRIC MEAN =";GM
460 PRINT:PRINT"HARMONIC MEAN =";HM
470 PRINT:PRINT"TOTAL DATA      =";N
480 PRINT:PRINT:PRINT"PRESS ANY KEY FOR REST OF ANALYSIS"
490 WHILE INKEY$="":WEND
500 M2=(TD-((TA^2)/N))/(N-1):SD=SQR(M2)
510 GOSUB 710
520 REM***** REST OF ANALYSIS *****
530 CLS
540 LOCATE 27,2:PRINT"REST OF ANALYSIS"
550 PRINT:PRINT"MEAN      =";AM;" + or -";SD
560 PRINT:PRINT"VARIANCE =";M2*(N-1)/N
570 LOCATE 1,12:PRINT"Kurtosis and Skewness (Y/N)"
580 A$=INKEY$
590 IF A$="N"OR A$="n"THEN END
600 IF A$<>"Y" AND A$<>"y"THEN 570
610 PRINT:PRINT"RE-ENTER DATA (9999 when done)"
620 IF GR=1 THEN INPUT DATUM;T ELSE INPUT"Group value";T:INPUT"Number of data";D
630 IF T=9999 THEN 660
640 DF=T-AM:M3=M3+DG*(DF^3):M4=M4+DG*(DF^4)
650 GOTO 620
660 SK=M3/(N*(M2^1.5))
670 KU=M4/(N*(M2^2))
680 PRINT:PRINT"SKEWNESS=";SK
690 PRINT:PRINT"KURTOSIS=";KU
700 END
710 CLS
720 SF=1
730 YLEN=(360\MOST)*MOST:IF YLEN=0 THEN MOST=MOST/10:SF=SF*10:GOTO 730
740 PLOT 40,32:DRAWR XLEN,0
750 PLOT 40,32:DRAWR 0,YLEN
760 LOCATE 1,8
770 FOR Z=1 TO 9
780 PRINT MID$("FREQUENCY",Z,1)
790 PRINT CHR$(13)+CHR$(8)
800 NEXT
810 MOVE 250,12:TAG:PRINT"DISTRIBUTION";
820 LOCATE 1,15
830 MOVE 28,38
840 PRINT"0";
850 MOVER -4,0
860 YDIV=YLEN/MOST
870 INC=INT(MOST/10.1)+1
880 FOR Z=1 TO 10
890 MOVER -(8+(LEN(STR$(Z*INC*SF))-2)*8),YDIV*INC
900 PRINT MID$(STR$(Z*INC*SF),2);
910 PLOTR 6,-6:MOVER -6,6
920 NEXT
930 XINC=INT(NUM/10.1)+1
940 CENT=-((INT(WIDE\2)-4)*(WIDE>29))
950 MOVE 32+CENT,26:PRINT LD;
960 FOR Z=1 TO 10
970 MOVE 40+WIDE*XINC*Z+CENT,26
980 PRINT MID$(STR$(LD+Z*XINC*DIV),2);
990 NEXT
1000 MOVE 41,32
1010 FOR P=1 TO NUM
1020 YLEN=C(P)*YDIV/SF
1030 FOR Q=1 TO WIDE+(WIDE>2)
1040 DRAWR 0,YLEN:MOVER 1,-YLEN
1050 NEXT Q
1060 MOVER -(WIDE>2),0
1070 NEXT P
1080 TAGOFF:LOCATE 26,1:PRINT"PRESS ANY KEY TO CONTINUE"
1090 WHILE INKEY$="":WEND
1100 RETURN

```